APPLICATIONS OF MAPLE IN THE PHYSICAL SCIENCES
Encounters with Chaos, Barry G. Adams
May 10, 2000 (Preliminary Copy)

1 Introduction

It was once thought that complex behaviour arose only from the study of complex systems. Recently and somewhat surprisingly it has been shown that even the simplest one-dimensional nonlinear dynamical systems can have unexpectedly complex behaviour leading to phenomena such as bifurcation, period doubling and chaos. In such systems small changes or inaccuracies in initial conditions can give rise to a completely different behaviour and predictions.

In this Chapter we first consider chaos in one dimensional maps using the logistic map as an example. Then the concept of a strange attractor is introduced using the two dimensional Henon map and the Lorenz system of three first order nonlinear differential equations.

2 One dimensional discrete dynamical systems

A one dimensional discrete dynamical system is a rule of the form

$$x_n = F(x_{n-1}),\ n = 1, 2, \ldots, \tag{1}$$

where $F : [a, b] \to R$ is real-valued function or mapping, whose range is contained in its domain (so the x_n are all defined). For each initial value x_0 the mapping (1) defines a sequence of iterates called the orbit of x_0 with respect to F and denoted by

$$\mathrm{Orb}_F(x_0) = [x_0, x_1, \ldots, x_{n-1}, x_n, \ldots]. \tag{2}$$

The study of a one dimensional discrete dynamical system, described by a map F, involves the characterization of these orbits. There are several possibilities:

1. *The orbit is periodic.* This means there is some smallest value of $n \ge 1$ such that $x_n = x_0$. The sequence of iterates repeats:

$$\begin{aligned} Orb_F(x_0) &= [x_0, \ldots, x_{n-1}, x_0, \ldots, x_{n-1}, \ldots], \\ &= \overline{x_0, \ldots, x_{n-1}}. \end{aligned}$$

   In this case we say that the orbit has period n. For $n = 1$ we say that the point x_0 is a fixed point since in this case $x_0 = F(x_0)$. The fixed points of F are the roots of the equation $x = F(x)$.

2. *The orbit is eventually periodic.* This means that there is some smallest value of $n \geq 1$ and some smallest value of $m \geq 1$ such that $x_{m+n} = x_m$ so the sequence has period n after some initial nonrepeating part:

$$\begin{aligned} Orb_F(x_0) &= [x_0, \ldots, x_{m-1}, x_m \ldots, x_{m+n-1}, x_m, \ldots, x_{m+n-1}, \ldots], \\ &= x_0, \ldots, x_{m-1}\overline{x_m \ldots, x_{m+n-1}}. \end{aligned}$$

   A periodic orbit is the special case with no initial nonrepeating part.

3. *The orbit is chaotic.* In this case there is no part that eventually repeats and the sequence appears to be random. It is usually very sensitive to initial conditions. A slight change in the initial value x_0 produces a very different sequence.

2.1 Maple procedure for calculating orbits

The following Maple procedure can be used to calculate an orbit of a map and display it in a convenient form. It is only necessary to supply the map `F`, the initial value `x0` defining the orbit, the number of iterations `skipiter` to skip before displaying the orbit (useful if the long term behaviour of the sequence is desired), and the number of rows in the array used to display the results. The iteration numbers and sequence values are displayed in a 6 column format for a compact display.

```
> maporbit := proc(F, x0, skipiter, displayiter)
>    local x, k, c, orbit;
>    x := x0; # starting value
>    # skip some iterations before collecting points
>    for k from 1 to skipiter do x := F(x); od:
>    orbit := array(1..displayiter,1..6);
>    for c from 1 to 3 do
>       for k from 1 to displayiter do
>          orbit[k,2*c-1] := skipiter+(c-1)*displayiter+k-1;
>          orbit[k,2*c] := x;
>          x := F(x);
>       od:
>    od:
>    op(orbit);
> end:
```

It is possible to convert the resulting array to a list for more convenient processing if desired.

```
> orbitlist := proc(orbit)
>    local i, j;
>    seq( seq( [orbit[i,2*j-1],orbit[i,2*j]],
>       i=1..linalg[rowdim](orbit)), j=1..3);
> end:
```

3 The logistic map

One of the icons of chaos theory is the one parameter logistic map defined by

$$Q_a(x) = ax(1-x),\ 0 \leq x \leq 1,\ 0 \leq a \leq 4 \tag{3}$$

Historically it had been assumed that complicated dynamical behaviour was the result of complex systems so it was a surprise that such a simple quadratic map could exhibit complex dynamical properties and could illustrate the period doubling route to chaos.

We can use procedure `maporbit` to calculate orbits for the logistic map. This procedure and the following applications of it are available in the worksheet `maporbit.ms`. For parameter value $a = 2.9$ we obtain the following results after skipping 1000 iterations.

```
> Q := (x) -> a*x*(1-x):
> a := 2.9: maporbit(Q, 0.1, 1000, 10);
```

1000	.6551724133	1010	.6551724133	1020	.6551724133
1001	.6551724144	1011	.6551724144	1021	.6551724144
1002	.6551724133	1012	.6551724133	1022	.6551724133
1003	.6551724144	1013	.6551724144	1023	.6551724144
1004	.6551724133	1014	.6551724133	1024	.6551724133
1005	.6551724144	1015	.6551724144	1025	.6551724144
1006	.6551724133	1016	.6551724133	1026	.6551724133
1007	.6551724144	1017	.6551724144	1027	.6551724144
1008	.6551724133	1018	.6551724133	1028	.6551724133
1009	.6551724144	1019	.6551724144	1029	.6551724144

This suggest that for $a = 2.9$ the orbit approaches a limit $x_\infty \approx 0.65551724$, a fixed point of the map Q. The orbit is eventually periodic with period 1.

■ **Exercise** The preceding example was done with the default 10 digits of numerical precision and it may appear from the data that the orbit is period 2 rather than period 1 since the values oscillate between two values. Try `Digits := 15` to convince yourself that $a = 2.9$ gives period 1. This shows that numerical results are not proofs. However it can be proven rigorously that for $0 < a \leq 3$ the period is 1.■

For parameter value $a = 3.2$ we obtain the following results after skipping 1000 iterations.

```
> a := 3.2: maporbit(Q, 0.1, 1000, 10);
```

1000	.5130445093	1010	.5130445093	1020	.5130445093
1001	.7994554906	1011	.7994554906	1021	.7994554906
1002	.5130445093	1012	.5130445093	1022	.5130445093
1003	.7994554906	1013	.7994554906	1023	.7994554906
1004	.5130445093	1014	.5130445093	1024	.5130445093
1005	.7994554906	1015	.7994554906	1025	.7994554906
1006	.5130445093	1016	.5130445093	1026	.5130445093
1007	.7994554906	1017	.7994554906	1027	.7994554906
1008	.5130445093	1018	.5130445093	1028	.5130445093
1009	.7994554906	1019	.7994554906	1029	.7994554906

Here we do not obtain a limit. Instead the orbit oscillates between the two values $p \approx 0.51304$ and $q \approx 0.79946$. The dynamics changes from a period 1 behaviour to a period 2 behaviour somewhere between $a = 2.9$ and $a = 3.2$. The actual parameter value is $a = 3$ and is called a bifurcation value. In fact as a increases through 3 an entire sequence of period doubling bifurcations takes place, ending in a region of chaos containing windows of periodic behaviour.

■ **Exercise** Experiment with the parameter values $a = 3.5$, $a = 3.56$, $a = 3.74$, $a = 3.84$. Interpret the results.■

Parameter value $a = 3.9$ produces the following orbit after skipping the first 10000 terms.

```
> a := 3.9: maporbit(Q, 0.1, 10000, 10);
```

10000	.3015539704	10010	.1644131389	10020	.7665540813
10001	.8214147763	10011	.5357876888	10021	.6979007948
10002	.5721009122	10012	.9700050410	10022	.8222575742
10003	.9547256882	10013	.1134715196	10023	.5699852178
10004	.1685757391	10014	.3923233619	10024	.9558980700
10005	.5466160412	10015	.9297823920	10025	.1644121041
10006	.9665250845	10016	.2546196726	10026	.5357849801
10007	.1261819476	10017	.7401751302	10027	.9700057971
10008	.4300142484	10018	.7500320367	10028	.1134687477
10009	.9558977790	10019	.7311875244	10029	.3923150049

It appears that there is no pattern here although we could be dealing with a very long period. One way to examine the difference in behaviour for $a = 3.2$ and $a = 3.9$ is to examine the sensitivity to the initial condition which we have always chosen to be $x_0 = 0.1$.

■ **Exercise** Repeat the calculations for $a = 3.2$ and $a = 3.9$ but using $x_0 = 0.100001$ as initial condition. Interpret the results.■

3.1 Time series for the logistic map

It also very easy to plot a time series for a one-dimensional map such as the logistic map using a line graph for the points (n, x_n). The iteration number n acts like a time variable. The following Maple procedure can be used to plot time series. This procedure and the following applications of it are available in a worksheet called `maptime.ms`.

```
> maptime := proc(F, x0, splist,caption)
>    local x, xlast, iter1, iter2, s, k, range, points, pts, p1, p2,
>          skipiter, plotiter;
>
>    # splist := [skip,plot,skip,plot,skip,plot,...]
>
>    x := x0:
>
>    iter1 := 0;
>    iter2 := 0;
>    for s from 1 to nops(splist)/2 do
>       skipiter := op(2*s-1,splist);
>       plotiter := op(2*s,splist);
>       iter1 := iter2 + skipiter;
>       iter2 := iter1 + plotiter;
>
>       # skip iterations
```

```
>
>       for k from 1 to skipiter do x := F(x); od:
>
>       # collect points for plotting iterations
>
>       points := array(1..plotiter+1);
>       for k from 1 to plotiter+1 do
>          points[k] := [iter1+k-1, x];
>          xlast := x;
>          x := F(x);
>       od:
>       pts := convert(points,list);
>       x := xlast; # so we begin next graph where we left off
>       range := iter1..iter2, 0..1;
>       p1 := plot(pts, range, color=RGB(0,0,0), style=LINE):
>       p2 := plot(pts, range, color=blue, symbol=BOX, style=POINT):
>       print(plots[display]([p1, p2],title=caption));
>    od;
> end:
```

The first parameter is the map, the second is the initial value of x, and the third parameter is a list of pairs of numbers. The first number of each pair is the number of iterations to skip and the second is the number of iterations to plot. For example, to plot the first 50 iterations use the list `[0,50]`. To skip the first 1000 iterations, plot the next 50 iterations, skip the next 1000 iterations and plot the next 30 iterations use the list `[1000,50,1000,30]`. This will produce two time series plots.

To illustrate the peridicity of the logistic map for the period 2 region and the period 3 window the following statements can be used to plot the initial part of the time series. The results are shown in Figure 1.

```
> Q := x -> a*x*(1-x):
> a := 2.9: maptime(Q, 0.1, [0,30,0,30],`a=2.9`);
> a := 3.2: maptime(Q,0.1,[0,50],`a=3.2`);
> a := 3.84: maptime(Q,0.1,[0,60],`a=3.84`);
```

For the parameter value $a = 3.9$, in the chaotic region, we can use

```
> a := 3.9: maptime(Q,0.1,[10000,100],`a=3.9`);
```

No periodicity is observed no matter how many iterations are skipped.

4 Graphical analysis of one dimensional maps

There is an interesting graphical interpretation of a one dimensional map F and its fixed points. Since the fixed points are solutions of $F(x) = x$ we can find them geometrically at the intersections of the line $y = x$ and the curve $y = F(x)$.

For the logistic map $Q_a(x)$ with $a = 2.9$ the following Maple statements produce the graph of the intersection of the curve $y = Q_a(x)$ and the line $y = x$ for parameter value $a = 3.9$.



Figure 1: Plots of time series for the logistic map $Q_a(x) = ax(1-x)$ illustrating convergence to a fixed point for $a < 3$ (top left), period 2 bifurcation (upper right), period 3 (bottom left) and chaos (bottom right).

```
> Q := (a,x) -> a*x*(1-x):
> OPTS := x=0..1, color=RGB(0,0,0), scaling=CONSTRAINED:
> diag := plot(x, OPTS):
> parabola := plot(Q(2.9,x), OPTS):
> plots[display]([diag,parabola]);
```

The following Maple statements give the fixed points and the nonzero fixed point for $a = 2.9$.

```
> [solve(x = Q(a,x), x)];
> subs(a=2.9,");
```

$$\left[0, -\frac{1-a}{a}\right]$$

$$[\,0, .6551724138\,]$$

4.1 Staircase and cobweb diagrams

We can also obtain a two dimensional graphical representation of an orbit, $\mathrm{Orb}_F(x_0)$, on the diagram by joining the points

$$(x_0, x_1),\ (x_1, x_1),\ (x_1, x_2),\ (x_2, x_2),\ \ldots,\ (x_n, x_{n+1}),\ \ldots,$$

which alternately lie on the curve $y = F(x)$ and the line $y = x$. This corresponds to repeating the sequence "move vertically to the curve $y = F(x)$ and then horizontally to the line $y = x$".

This graphical interpretation of an orbit gives the so-called staircase diagram in case there is monotone convergence to a fixed point or the cobweb diagram in case there is oscillatory convergence to a period n point for $n \geq 1$. The results of this subsection are available in the worksheets `griter.ms` and `mapiter.ms`.

To illustrate the staircase diagram consider the logistic map with $a = 1.5$ and $x_0 = 0.1$ and the Maple statements

```
> Q := (a,x) -> a*x*(1-x):
> a := 1.5: n := 10: x[0] := 0.1:
> for i from 1 to n do x[i] := Q(a,x[i-1]); od:
> p := seq( op([[x[i-1],x[i]], [x[i],x[i]]]),i=1..n);
```

$$\begin{aligned} p := {} & [.1, .135], [.135, .135], [.135, .1751625], [.1751625, .1751625],\\ & [.1751625, .2167208979], [.2167208979, .2167208979],\\ & [.2167208979, .2546294255], [.2546294255, .2546294255],\\ & [.2546294255, .2846899218], [.2846899218, .2846899218],\\ & [.2846899218, .3054623553], [.3054623553, .3054623553],\\ & [.3054623553, .3182326572], [.3182326572, .3182326572],\\ & [.3182326572, .3254409496], [.3254409496, .3254409496],\\ & [.3254409496, .3292937069], [.3292937069, .3292937069],\\ & [.3292937069, .3312890423], [.3312890423, .3312890423] \end{aligned}$$

```
> OPTS := x=0.0..0.5, color=RGB(0,0,0):
> diag := plot(x, OPTS):
> stair := plot([p],OPTS):
> parabola := plot(Q(a,x),OPTS):
> plots[display]([diag,parabola,stair],scaling=CONSTRAINED);
```

Here we get monotone convergence to the fixed point 1/3 as shown in the left plot of Figure 2.

To illustrate the cobweb diagram consider the parameter value $a = 2.9$ and the Maple statements.

```
> a := 2.9: n := 10: x[0] := 0.1:
> for i from 1 to n do x[i] := Q(a,x[i-1]); od:
> p := seq( op([[x[i-1],x[i]], [x[i],x[i]]]),i=1..n);
```

$$\begin{aligned} p := {} & [.1, .261], [.261, .261], [.261, .5593491], [.5593491, .5593491],\\ & [.5593491, .7147852846], [.7147852846, .7147852846],\\ & [.7147852846, .5912151163], [.5912151163, .5912151163], \end{aligned}$$

$$
\begin{aligned}
&[\,.5912151163, .7008714273\,], [\,.7008714273, .7008714273\,],\\
&[\,.7008714273, .6079869421\,], [\,.6079869421, .6079869421\,],\\
&[\,.6079869421, .6911825789\,], [\,.6911825789, .6911825789\,],\\
&[\,.6911825789, .6190027425\,], [\,.6190027425, .6190027425\,],\\
&[\,.6190027425, .6839312070\,], [\,.6839312070, .6839312070\,],\\
&[\,.6839312070, .6268910021\,], [\,.6268910021, .6268910021\,]
\end{aligned}
$$

```
> OPTS := x=0.0..1.0, color=RGB(0,0,0):
> diag := plot(x, OPTS):
> cob := plot([p],OPTS):
> parabola := plot(Q(a,x),OPTS):
> plots[display]([diag,parabola,cob],scaling=CONSTRAINED);
```

Here we get oscillatory convergence to the fixed point as shown in the right plot of Figure 2.



Figure 2: Staircase and cobweb diagrams for the logistics map for $a = 1.5$ on the left and $a = 2.9$ on the right

Here is a general procedure for the graphical analysis of a one dimensional map.

```
> mapiter := proc(F, x0, a, b, skipiter, plotiter, caption)
>    local xp, k, p, orbit, curve, diag;
>
>    xp := x0; # starting value
>
>    # Skip the first few iterations if long term behaviour is desired.
>    # To plot all iterations use 0 for skipiter.
>
>    p := array(0..plotiter);
>    for k from 1 to skipiter do xp := F(xp); od:
>    for k from 0 to plotiter do p[k] := xp; xp := F(xp); od;
>
>    curve := plot(F(x), x=a..b, color=blue);
>    diag := plot(x, x=a..b, color=RGB(0,0,0));
>    orbit := plot([seq(op([[p[k-1],p[k]],[p[k],p[k]]]), k=1..plotiter)],
>             color=red);
>    plots[display]([orbit, curve, diag],title=caption,scaling=CONSTRAINED);
> end:
```

The following statements use `mapiter` to do the graphical analysis for $a = 3.84$, in the period 3 window and $a = 3.9$ in the chaotic region.

```
> Q := (x) -> a*x*(1-x):
> a := 3.84: mapiter(Q, 0.1, 0, 1, 1000, 100,`a=3.84`); # period 3
> a := 3.9:  mapiter(Q, 0.1, 0, 1, 1000, 100,`a=3.9`); # chaos
```

The results are shown in Figure 3.



Figure 3: Graphical iteration of the logistic map from `mapiter` for $a = 3.84$ in the period 3 window (left) and $a = 3.9$ in the chaotic region (right).

■ **Exercise** Perform a graphical analysis for other values of a such as $a = 3.2$, $a = 3.5$, $a = 3.56$ and $a = 3.74$.■

■ **Exercise** Use `mapiter` for $F(x) = \cos(x)$, $F(x) = e^{-x}$ and $F(x) = \text{Arctan}(x)$.■

5 Strange Attractors

Another feature of chaos theory which can be illustrated with Maple is the strange attractor which first appears in the study of two dimensional discrete dynamical systems specified by rules of the form

$$x_n = F(x_n, y_n) \quad (4)$$
$$y_n = G(x_n, y_n) \quad (5)$$

5.1 The Henon Attractor

We consider the Henon attractor which arises from the two parameter mapping defined by

$$F(x, y) = 1 - ax^2 + y, \qquad G(x, y) = bx \quad (6)$$

The standard parameter values $a = 1.4$ and $b = 0.3$ can be used to illustrate the concept of a strange attractor. The following pseudocode algorithm can be used to discover the attractor.

$x \leftarrow x_0$: $y \leftarrow y_0$
for $k \leftarrow 1$ to *maxiter*
  $x_t \leftarrow 1 - ax^2 + y$
  $x \leftarrow x_t$
  $y \leftarrow bx_t$
  if $k >$ *skipiter* then "plot point at (x, y)" endif
endfor

Thus, except for the first few points, we plot the points in the orbit. The picture that "develops" is called the Henon attractor. The orbit points wander around the attractor in a random fashion. The orbits are very sensitive to the initial conditions, a sign of chaos, but the attractor appears to be a stable geometrical object that is not sensitive to initial conditions.

Unfortunately the above algorithm is not easily implemented in Maple since the large number of points needed must be stored in a large array or list before any plotting takes place. Thus we need another way to "develop" the attractor.

In his analysis of this map Henon defined a trapping quadrilateral and showed that all points on and inside this quadrilateral did not escape to infinity as they were iterated. Instead they remained inside the quadrilateral forever. At each iteration the quadrilateral is stretched and folded by the Henon map until the geometrical attractor is obtained. This gives a nice example of the stretching and folding interpretation, or baker's model, of chaos.

The following Maple statements define this quadrilateral as 4 sides in parametric form and plot it.

```
> line := [xa + (xb-xa)*t, ya + (yb-ya)*t]:
> v := [[-1.33,0.42], [1.32,0.133], [1.245,-0.14], [-1.06,-0.5], [-1.33,0.42]]:
> quad[0] := [seq(subs(xa=v[s][1],xb=v[s+1][1],
>                 ya=v[s][2],yb=v[s+1][2],line), s=1..4 )];
```

$$quad_0 := [[-1.33 + 2.65\,t, .42 - .287\,t], [1.32 - .075\,t, .133 - .273\,t], \\ [1.245 - 2.305\,t, -.14 - .36\,t], [-1.06 - .27\,t, -.5 + .92\,t]]$$

```
> OPTS := -1.5..1.5, -0.5..0.5, color=RGB(0,0,0):
> quadrilateral := plot(v,OPTS):
> #interface(plotdevice=postscript, plotoutput=`henon0.ps`);
> plots[display](quadrilateral,title=`Iteration 0`);
```

Now we can iterate these parametric equations to obtain the iterated curves in parametric form.

```
> a := 1.4:
> b := 0.3:
> kmax := 6:
> Henon := [1 - a*x^2 + y, b*x]:
> for k from 1 to kmax do
>    quad[k] := [seq(subs(x=quad[k-1][s][1],y=quad[k-1][s][2],Henon),s=1..4)];
> od:
```

Finally we can make a sequence of plot structures for each iteration. Since the length of the curve approximately doubles at each iteration, due to the stretching and folding, it is necessary to double the number of points used to render the parametric curves. We show the results for iterations 0 and 1 in Figure 4 and for iterations 5 and 6 in Figure 5. There is little difference between iterations 5 and 6.

```
> points := [50,100,200,600,1200,2400]:
> for k from 1 to kmax do
>    caption := cat(`Iteration `,k);
>    p := seq(plot([op(quad[k][s]),t=0..1],
>          OPTS,numpoints=points[k]), s=1..4):
>    if k=1 or k=kmax then
>       print(plots[display]([quadrilateral,p],title=caption));
>    fi;
> od:
```



Figure 4: Iterations 0 and 1 of the Henon map.



Figure 5: Iterations 5 and 6 of the Henon map.

5.2 The Lorenz Attractor

Strange attractors also arise in the solution of systems of first order differential equations. For autonomous systems (no explicit time dependence) the minimum number of equations needed is three and the classic example, another icon of chaos theory, is the Lorenz attractor arising from the system

$$\frac{dx}{dt} = \sigma(-x+y) \quad (7)$$

$$\frac{dy}{dt} = rx - y - xz \quad (8)$$

$$\frac{dz}{dt} = -bz + xy \quad (9)$$

Lorenz fixed $\sigma = 10$, $b = 8/3$ and followed the solution trajectories in three dimensional phase space for various values of the remaining parameter r. Each trajectory is a curve in this space parametrized by the points $(x(t), y(t), z(t))$. For some values of r the trajectories approached well-defined periodic orbits and for other values of r the trajectories approached a strange butterfly shaped attractor and wandered around it in a random fashion.

The results of this section are available in the worksheet `lorenz.ms`. The following procedure can be used to analyze the trajectories using the Maple `dsolve` procedure with the `numeric` option.

```
> lorenz := proc(sigma, b, r, ts2, te2, ts1, te1, steps, fnameprefix)
>    local de1, de2, de3, desys, g, k, s, T, caption, xt,
>       dim, tp, xp, yp, zp, tt, dt;
>
>    de1 := diff(x(t),t) = sigma*(-x(t)+y(t));
>    de2 := diff(y(t),t) = r*x(t) - y(t) - x(t)*z(t);
>    de3 := diff(z(t),t) = -b*z(t) + x(t)*y(t);
>    desys := {de1,de2,de3,x(0)=1,y(0)=1,z(0)=1}, [x(t),y(t),z(t)];
>
>    dim := steps;
>    tp := array(1..dim);
>    xp := array(1..dim);
>    yp := array(1..dim);
>    zp := array(1..dim);
>
>    # Solve the system and save results in arrays for plotting
>    # only save long term behaviour.
>
>    k := 0;
>    dt := (te2-ts2)/(steps-1);
>    g := dsolve(desys, type=numeric,'maxfun'=-1);
>    for tt from ts2 to te2 by dt do
>       s := g(tt);
>       k := k + 1;
>       tp[k] := tt;
>       xp[k] := rhs(s[2]); yp[k] := rhs(s[3]); zp[k] := rhs(s[4]);
>    od;
>    dim := k;
>
>    # plot some projections of phase space trajectories
>
>    caption := cat(`r = `,convert(r,string));
>    T := title = caption;
>    interface(plotoutput=fnameprefix.`xy.ps`);
>    print(plot( [seq( [xp[k],yp[k]], k=1..dim)], labels=[`x`,`y`],T ));
>    interface(plotoutput=fnameprefix.`xz.ps`);
>    print(plot( [seq( [xp[k],zp[k]], k=1..dim)], labels=[`x`,`z`],T ));
```

```
>     interface(plotoutput=fnameprefix.`yz.ps`);
>     print(plot( [seq( [yp[k],zp[k]], k=1..dim)], labels=[`y`,`z`],T ));
>     interface(plotoutput=fnameprefix.`xyz.ps`);
>     print(plots[spacecurve]([seq( [xp[k],yp[k],zp[k]], k=1..dim)],
>           axes=NORMAL,labels=[`x`,`y`,`z`],shading=NONE,T));
>
>     # resolve the system on the larger interval ts1<=t<=te1
>     # to graph x(t) against t
>
>     g := dsolve(desys, type=numeric,'maxfun=-1');
>     interface(plotoutput=fnameprefix.`tx.ps`);
>     xt := plots[odeplot](g, [t,x(t)], ts1..te1,
>        numpoints=2000,title = cat(caption,` (x versus t)`));
>     print(plots[display]([xt],labels=[`t`,`x`]));
> end:
```

An example of a nonsymmetric periodic orbit can be obtained for $r = 225$ using

```
> interface(plotdevice=postscript);
> lorenz(10, 8/3, 225.0, 3.0, 3.5, 0.0, 5.0, 501,`L225`);
```

Some results are shown in Figure 6.



Figure 6: Lorenz trajectories for $r = 225$: projection in the xy plane (top left), the xz plane (top right), space curve in xyz space (bottom left) and time series for $x(t)$ (bottom right).

An example of symmetric periodic orbit can be obtained for $r = 475$ using

```
> lorenz(10, 8/3, 475.0, 3.0, 3.34, 0.0, 5.0, 501, ‘L475‘);
```

Some results are shown in Figure 7.



Figure 7: Lorenz trajectories for $r = 475$: projection in the xy plane (top left), the xz plane (top right), space curve in xyz space (bottom left) and time series for $x(t)$ (bottom right).

A period doubling sequence of symmetric periodic orbits can be obtained using $r = 475$, $r = 163.5$, $r = 93$, $r = 69.75$. Some projections for $r = 93$ can be obtained from

```
> lorenz(10, 8/3, 93.0, 10.0, 12.5, 0.0, 13.0, 1001, ‘L93‘);
```

Some results are shown in Figure 8.

To illustrate chaotic trajectories which are very sensitive to initial conditions we can try $r = 28$ and

```
> lorenz(10, 8/3, 28.0, 10.0, 40.0, 0.0, 30.0, 3001, ‘L28‘);
```

Some results are shown in Figure 9



Figure 8: Lorenz trajectories for $r = 93$: projection in the xy plane (top left), the xz plane (top right), space curve in xyz space (bottom left) and time series for $x(t)$ (bottom right).



Figure 9: Lorenz trajectories for $r = 28$: projection in the xy plane (top left), the xz plane (top right), space curve in xyz space (bottom left) and time series for $x(t)$ (bottom right).

References

[DEVA89] Robert L. Devaney, “An Introduction to Chaotic Dynamical Systems”, 2nd Ed., Addison-Wesley, Menlo Park, Califoenia, 1989.

[FROY84] Jan Froyland and Knut H. Alfsen, “Lyapunov-exponent spectra for the Lorenz model”, Phys. Rev. **A29**, pp. 2929–2931, 1984.

[FROY92] Jan Froyland, “Introduction to Chaos and Coherence”, Institute of Physics Publishing, Bristol, (1992)

[GULI92] Denny Gulick, “Enocunters with Chaos”, McGraw-Hill, New York, 1992.

[HENO76] Michel Henon, “A Two-Dimensional Mapping with a Strange Attractor”, Commun. Math. Phys. **5D**, pp. 66-77, 1976.

[LORE63] Edward N. Lorenz, “Deterministic Nonperiodic Flow”, J. Atmospheric Sciences **20**, pp. 130–141, 1963

[MAY76] Robert M. May, “Simple Mathematical Models with Very Compicated Dynamics”, Nature **261**, pp. 459–467, 1976.

[PEIT92] Heinz-Otto Peitgen, Hartmut Jürgens and Dietmar Saupe, “Chaos and Fractals: New Frontiers of Science”, Springer-Verlag, New York, 1992.